REFFERENCES

1. Reid, J.L. "Deep Ocean Circulation." in Oceanography: The Last Frontier, Forum Series (Vetter, R.C. ed.) PP. 225-239, Voice of America, 1974.

2. Sverdrup, H.U., Johnson, M.W. and Fleming, R.H. The Oceans. Modern Asia ed., pp. 489-491, Prentice-Hall, N.J., 1975.

3. Pierson, W.J. "Waves." in Oceanography: The Last Frontier, Forun Series (Vetter, R.C. ed.) pp. 185-196, Voice of America, 1974.

4. von Arx, W.S. An Introduction to Physical Oceanography, pp. 113-116, Addison-Wesley Publishing, U.S.A., 1967.

5. Defant, A. Vol. I of Physical Oceanography, pp. 398-436, Pergamon Press, London, 1961.

6. Pond, S. and Pickard, G.L. Introductory Dynamic Oceanography, pp. 81-91, Peragmon Press, London, 1978.

7. Veronis, G. "Dynamics of Large-Scale Ocean Circulation" in Evolution of Physical Oceanography p. 147, Halliday Lithograph, U.S.A., 1981.

8. Neelasri, C. "The Mean Flow in the Upper Half of the Gulf of Thailand During a Transition Period of the Monsoon." in Proceeding of the 2nd Seminar on the Water Quality and the Quality of Living Resources in Thai Water, pp. 57-63, Bangkok, May 26-28, 1981.

9. Jiraporn, J. and Jia, P. "A Model Study of Wind Driven Circulation in the Upper Gulf of Thailand" Bangkok: Department of Marine Science, Chulalongkorn University, 1981 (Mimeographed).

10. Jiraporn, J. "The Orbital Flow Theory." Bangkok: Department of Marine Science, Chulalongkorn University, 1982 (Mimeographed).

11. Neumann, G. and Pierson, W.J. Principle of Physical Oceanography, pp. 191-194, Prentice-Hall, N.J., 1966.

12. Gross, M. G. Oceanography: A View of the Earth 2nd ed., pp. 222-227, Prentice-Hall, N.J., 1972.

13. Hydrographic Department, Maritime Safety Agency "Physical Oceanography Survey." Japan International Cooperation Agency, Japan, 1980.

14. Muga, B.J. and Wilson, J.F. Dynamic Analysis of Ocean Structures, p 31, Plenum Prews, N.Y., 1970.

15. Silvester, R. Coastal Engineering Vol. I, pp. 160-163, Elsevier Scientific Publishing, Amsterdam, 1974.

16. Yavorsky, B. and Detlaf, A. Handbook of Physics 2nd ed., pp. 214-215, Mir Publishers, Moscow, 1975.

17. Allen, J.S. "Models of Wind-Driven Currents on the Continental Shelf." Ann. Rev. of Fluid Mech., 12(1980): 389-433.

18. Lee, D.S. and Renouard, D.P. "Experimental Study of the Coastal Thermocline Height Variations Caused by an Impulsively Applied Wind of Limited Along-Shore Extent". Dyn. of Atmos. and Oceans 7(1983): 129-145.

APPENDIX A

EXAMPLE OF DATA ANALYSIS

In each experiment, drifter of five different depths (0 cm, 3 cm, 5 cm, 7 cm and 13 cm) were used for observation of induced current. The numbers of drifter for each depht varied.

Table A shows the data and calculated velocities for experiment 6. The positions of a drifter on the T.V. monitor were marked every 1.4 seconds. The origin of the coordinate system used for depicting the position of a drifter was arbitrarily set up. Fig. A shows the orientation of coordinate systems.

Velocity components, u and v were calculated from the following formula

$$u = \frac{x_{i+1} - x_i}{\Delta t} s_x \qquad (A.1)$$

$$v = \frac{y_{i+1} - y_i}{\Delta t} s_y \qquad (A.2)$$

where x_i (y_i) and x_{i+1} (y_{i+1}) are the x (y) ordinates of the successive position of a drifter. Δt is time interval of 1.4 seconds. S_x and S_y are the ratios of the actual range dimension to the range dimension on the T.V. monitor. These two ratios are unequal because of the vedio distortion. The values of S_x and S_y are as follow

s_x = 1.85

s_y = 1.92

<u> and <v> are the average velocity of each drifter. $\bar{u}$ and $\bar{v}$ are the average velocity of all drifters of the same depth.

Table A Velocity Data form Experiment 6

z	Float No	Co-ordinate (x , y)	x - component u	<u>	$\bar{u}$	y - component v	<v>	$\bar{v}$
0	1	(-7.9, 4.6)						
		(-6.1, 7.1)	2.3			3.4		
		(-3.3, 9.5)	3.7	3.0		3.3	3.4	
	2	(-2.3, 0.9)						
		(-0.7, 3.7)	2.1			3.9		
		(1.6, 4.8)	3.0	2.6		1.5	2.7	
	3	(-5.7, 2.6)						
		(-3.5, 3.4)	2.9			1.1		
		(-3.2, 5.1)	0.4			2.3		
		(-1.4, 5.6)	2.4	1.9		0.7	1.4	
	4	(-4.7, 3.3)						
		(-2.5, 6.2)	2.9			3.9		
		(-1.4, 9.3)	1.5	2.2	2.4	4.3	4.1	2.9

(cont.)

z	Float No	Co-ordinate (x , y)	x - component u	<u>	ū	y - component v	<v>	v̄
3	1	(-7.2, 6.3)						
		(-6.3, 6.3)	1.2			0.0		
		(-5.5, 5.9)	1.1			-0.6		
		(-4.8, 5.3)	0.9	1.1		-0.8	-0.5	
	2	(-1.7, 7.0)						
		(-0.5, 7.8)	1.6			1.1		
		(0.5, 7.5)	1.3	1.5	1.3	-0.4	0.3	-0.1
5	1	(-2.9, 3.2)						
		(-2.9, 4.2)	0.0			1.4		
		(-2.6, 5.3)	0.4			1.5		
		(-2.1, 6.0)	0.7			0.9		
		(-2.1, 6.6)	0.0	0.3		0.8	1.2	

(cont.)

z	Float No	Co-ordinate (x , y)	x - component u	<u>	$\bar{u}$	y - component v	<v>	$\bar{v}$
	2	(-6.5, 4.1)						
		(-6.8, 5.5)	-0.4			1.9		
		(-2.5, 6.6)	0.4			1.5		
		(-2.0, 7.2)	0.7			0.8		
		(-1.1, 7.5)	1.2	0.5	0.4	0.4	1.2	1.2
7	1	(-4.0, 5.4)						
		(-3.5, 3.9)	0.7			0.7		
		(-3.3, 6.4)	0.3			0.7		
		(-3.2, 9.8)	0.1	0.4		0.6	0.7	
	2	(-7.3, 4.0)						
		(-6.4, 4.7)	1.2			0.9		
		(-5.8, 5.4)	0.8			0.9		
		(-4.7, 5.6)	1.5	1.2		0.3	0.7	
	3	(-2.5, 4.2)						
		(-1.0, 5.9)	1.9			2.3		
		(-0.2, 8.5)	1.1	1.5	1.0	3.6	2.9	1.4

(cont.)

z	Float No	Co-ordinate (x , y)	x - component u	<u>	$\bar{u}$	y - component v	<v>	$\bar{v}$
13	1	(-0.7, 2.3)						
		(-1.7, 3.3)	-1.3			1.4		
		(-2.7, 4.5)	-1.3			1.7		
		(-3.6, 5.0)	-1.2	-1.3		0.7	1.2	
	2	(-2.3, 3.8)						
		(-3.2, 4.6)	-1.2			1.1		
		(-3.8, 5.8)	-0.8			1.7		
		(-4.2, 6.6)	-0.5	-0.8		1.1	1.3	
	3	(-4.0, 5.3)						
		(-4.9, 4.0)	-1.2			-1.8		
		(-5.8, 4.4)	-1.2			0.6		
		(-6.5, 3.9)	-0.9	-1.1	-1.1	-0.7	-0.6	0.6



Figure A.1 Arbitrary xy-axis Drawn on T.V. Screen

APPENDIX B

LIST OF COMPUTER PROGRAMS

```
1  REM   ********************
2  REM   *                  *
3  REM   * PROGRAM TO INPUT *
4  REM   * OBSERVATION DATA *
5  REM   *                  *
6  REM   * APPLESOFT  BASIC *
7  REM   *                  *
8  REM   ********************

10  REM   PROGRAM RUN HERE
12  GOSUB 100
13  REM  ASK FILE
14  GOSUB 200
15  REM  INPUT PARAMETERS
16  GOSUB 300
17  REM  INPUT DATAS
18  GOSUB 400
19  REM  FINAL FLAG
20  REM

100  REM   SUBROUTINE TO ASK FILE NAME
101  HOME
102  VTAB 3
103  INVERSE
104  HTAB 17: PRINT "       "
105  HTAB 17: PRINT " INPUT "
106  HTAB 17: PRINT "       "
107  NORMAL
108  VTAB 13
109  PRINT "FILE NO. ";
110  INPUT " >>> ";N$
111  GOSUB 520
112 N$ = "DATA " + N$
113  PRINT "FILE NAME IS ";
114  INVERSE
115  PRINT " ";N$;" "
116  NORMAL
117  GOSUB 510
118  RETURN
119  REM
```

```
200  REM SUBROUTINE TO INPUT PARAMETERS
201  HOME
202  INPUT "TIME DURATION  ";DT
203  INPUT "ROT. PERIOD    ";OM
204  INPUT "WAVE AMP.      ";AM
205  INPUT "WAVE LENGTH    ";LA
206  INPUT "WAVE FREQ.     ";FR
207  PRINT : PRINT
208  PRINT "KEEP?  (Y/N)  ";
209  GET Q$
210  IF Q$ = "N" THEN 200
211  HOME : INVERSE
212 Q$ = "CREATE " + N$ + " FILE"
213  VTAB 13: HTAB  LEN (Q$) / 2
214  PRINT "  ";Q$;"  ": NORMAL
215  GOSUB 540
216  RETURN
217  REM


300  REM  SUBROUTINE TO INPUT DATA
301  FOR I = 1 TO 5
302  HOME : PRINT " DEPTH # ";I: PRINT
303  INPUT "DRIFTER DEPTH  = ";Z(I): PRINT
304  INPUT "# OF DRIFTER   = ";NF(I): PRINT
305  PRINT : PRINT : PRINT "KEEP?   (Y/N)";
306  GET Q$: PRINT
307  IF Q$ = "N" THEN 302
308  GOSUB 560: REM

309  FOR J = 1 TO NF(I)
310  HOME
311  PRINT "DRIFTER DEPTH = ";Z(I)
312  PRINT "(THERE ARE ";NF(I);" DRIFTERS)"
313  PRINT : INVERSE
314  PRINT : PRINT "DRIFTER # ";J;" "
315  NORMAL : PRINT
316  GOSUB 530: GOSUB 530
317  INPUT " # OF DATAPAIRS ";ND(I,J): PRINT
318  IF ND(I,J) > 10 THEN 317
319  FOR K = 1 TO ND(I,J)
320  INPUT "X ";X(I,J,K)
321  INPUT "Y ";Y(I,J,K)
322  PRINT : NEXT K: REM
```

```
323  REM  CHECK DATA
324  HOME
325  INVERSE : PRINT "  CHECK DATAS  "
326  NORMAL : PRINT
327  PRINT "DEPTH = ";Z(I)
328  PRINT "# OF BUOY = ";NF(I)
329  PRINT "-------------"
330  PRINT "BUOY # ";J: PRINT
331  FOR K = 1 TO ND(I,J)
332  PRINT "(";X(I,J,K);" , ";Y(I,J,K);")"
333  NEXT K
334  PRINT : PRINT : PRINT
335  PRINT "--------------"
336  PRINT : PRINT : PRINT "KEEP?  (Y/N)";
337  GET Q$: PRINT
338  IF Q$ = "N" THEN 340
339  GOTO 351
340  PRINT : PRINT "1 >> CHANGE ND","2 >> CHANGE X,Y"
341  GET Q$
342  IF Q$ = "1" THEN 310
343  INPUT "HOW MANY PAIRS? ";KC
344  FOR CK = 1 TO KC
345  INPUT "WHICH PAIR?  ";K
346  INPUT " X   ";X(I,J,K)
347  INPUT " Y   ";Y(I,J,K)
348  NEXT CK
349  GOTO 323
350  REM

351  GOSUB 570
352  NEXT J: NEXT I
353  RETURN
354  REM


400  REM   SUBROUTINE TO SET FINAL FLAG
401  REM  FINAL FLAG
402  REM
403  HOME
404  VTAB 13: HTAB 18
405  FLASH
406  PRINT "FINAL"
407  NORMAL
408  FOR T = 1 TO 1000: NEXT T
409  VTAB 20
410  PRINT "1 >> RUN AGAIN"
411  PRINT "2 >> END"
412  GET Q$
413  IF Q$ = "1" THEN 10
414  IF Q$ = "2" THEN  END
415  GOTO 412
416  RETURN
417  REM
```

```
500  REM  ------------------------
501  REM  MISCELLANEOUS SUBROUTINES
502  REM  ------------------------

510  REM   DELAY TIME
511  FOR T = 1 TO 1000: NEXT T
512  RETURN
513  REM

520  REM   PRINT LINE
521  PRINT
522  FOR P = 1 TO 40
523  PRINT "-";
524  NEXT P
525  PRINT
526  RETURN
527  REM

530  REM   GENERATE SOUND
531 A =  - 16336
532  FOR S = 1 TO 20
533 SO =  PEEK (A)
534  NEXT S
535  RETURN
536  REM

540  REM   CREATE FILE
541 Y$ =  CHR$ (4)
542  PRINT Y$;"OPEN";N$
543  PRINT Y$;"DELETE";N$
544  PRINT Y$;"OPEN";N$
545  PRINT Y$;"WRITE";N$
546  PRINT DT: PRINT DM
547  PRINT FR: PRINT AM
548  PRINT LA: PRINT "DATA"
549  PRINT Y$;"CLOSE";N$
550  RETURN
551  REM

560  REM   APPEND Z & NF
561  PRINT Y$;"APPEND";N$
562  PRINT Y$;"WRITE";N$
563  PRINT Z(I): PRINT NF(I)
564  PRINT Y$;"CLOSE";N$
565  RETURN
566  REM

570  REM   APPEND COORDINATES
571 Y$ =  CHR$ (4)
572  PRINT Y$;"APPEND";N$
573  PRINT Y$;"WRITE";N$
574  PRINT " ND ": PRINT ND(I,J)
575  FOR K = 1 TO ND(I,J)
576  PRINT X(I,J,K): PRINT Y(I,J,K)
577  NEXT K
578  PRINT Y$;"CLOSE";N$
579  RETURN
580  REM
```

```
1  REM ************************
2  REM *                      *
3  REM *  PROGRAM TO FIND     *
4  REM *                      *
5  REM *      VELOCITY        *
6  REM *         &&           *
7  REM * STANDARD  DEVIATION  *
8  REM *                      *
9  REM ************************


10  REM
11 : FOR N = 1 TO 9
12 :: GOSUB 100: REM  LOAD DATA
13 :: GOSUB 200: REM  EVALUATE
14 :: GOSUB 300: REM  PRINT
15 : NEXT N
16 : END
17  REM


100  REM  SUBROUTINE TO LOAD DATA
101 N$ =  STR$ (N):D$ = "DATA " + N$
102  HOME : VTAB 10
103  PRINT "  LOADING  >>   ";
104  HTAB 20
105  INVERSE : PRINT D$: NORMAL : REM

106  REM  LOAD RESULT
107 Y$ =  CHR$ (4)
108  PRINT Y$;"OPEN";D$;",D2"
109  PRINT Y$;"READ";D$
110  INPUT DT: INPUT OM
111  INPUT FR: INPUT AM
112  INPUT LA: INPUT X$
113  FOR I = 1 TO 5
114  INPUT Z(I): INPUT NF(I)
115  FOR J = 1 TO NF(I)
116  INPUT Z$: INPUT ND(I,J)
117  FOR K = 1 TO ND(I,J)
118  INPUT X(I,J,K): INPUT Y(I,J,K)
119  NEXT K,J,I
120  PRINT Y$;"CLOSE";D$
121  RETURN
122  REM
```

```
200  REM  SUBROUTINE TO CALCULATE
201  VTAB 12: FLASH
202  PRINT "  CALCULATION  ": NORMAL
203  REM
204 SX = 5 / 2.6:SY = 5 / 2.7
205  FOR I = 1 TO 5
206  FOR J = 1 TO NF(I)
207  FOR K = 1 TO ND(I,J) - 1
208 X1 = X(I,J,K + 1)
209 X0 = X(I,J,K)
210 Y1 = Y(I,J,K + 1)
211 Y0 = Y(I,J,K)
212 DX = X1 - X0
213 DY = Y1 - Y0
214 VX(I,J,K) = DX * SX / 1.4
215 VY(I,J,K) = DY * SY / 1.4
216  NEXT K,J,I
217  REM
218  REM  AV.VELOCITY
219  FOR I = 1 TO 5
220  FOR J = 1 TO NF(I)
221 TX = 0:TY = 0
222 N1 = ND(I,J) - 1
223  FOR K = 1 TO N1
224 TX = VX(I,J,K) + TX
225 TY = VY(I,J,K) + TY
226  NEXT K
227 AXV(I,J) = TX / N1
228 AYV(I,J) = TY / N1
229  NEXT J,I
230  REM
231  REM  MEAN VELOCITY
232  FOR I = 1 TO 5
233 TX = 0:TY = 0
234  FOR J = 1 TO NF(I)
235 TX = TX + AXV(I,J)
236 TY = TY + AYV(I,J)
237  NEXT J
238 MXV(I) = TX / NF(I)
239 MYV(I) = TY / NF(I)
240  NEXT I
241  REM

242  REM  STANDARD DEVIATION
243  FOR I = 1 TO 5
244  FOR J = 1 TO NF(I)
245 XX = (AXV(I,J) - MXV(I)) ^ 2
246 YY = (AYV(I,J) - MYV(I)) ^ 2
247 XSUM = XSUM + XX
248 YSUM = YSUM + YY
249  NEXT J
250 XD = XSUM / (NF(I) - 1)
251 XSD(I) =  SQR (XD)
252 YD = YSUM / (NF(I) - 1)
253 YSD(I) =  SQR (YD)
254 XSUM = 0:YSUM = 0
255  NEXT I
256  RETURN
257  REM
```

```
300  REM  SUBROUTINE TO PRINT
301  HOME
302  PRINT  CHR$ (14);D$
303  GOSUB 420: HTAB 14
304  PRINT "PARAMETERS"
305  PRINT "TANK PERIOD",
306  PRINT OM; TAB( 27)"S"
307  PRINT "WAVE FREQ",
308  PRINT FR; TAB( 27)"CPS"
309  PRINT "WAVE LNGTH",
310  PRINT LA; TAB( 27)"CM"
311  PRINT "WAVE AMP",
312  PRINT AM; TAB( 27)"CM"
313  GOSUB 420
314  FOR I = 1 TO 5: GET W$
315  PRINT "BUOY DEPTH = ";
316  PRINT Z(I);" CM": PRINT
317  PRINT "FLOAT #"; TAB( 12)"AV(Y-COMP)"; TAB( 28)"AU(X-COMP"
318  FOR J = 1 TO NF(I)
319 RD = AXV(I,J): GOSUB 410:AX = RD
320 RD = AYV(I,J): GOSUB 410:AY = RD
321  PRINT  TAB( 3)J; TAB( 12)AX; TAB( 28)AY
322  NEXT J: GOSUB 420: NEXT I
323  GET W$
324  REM
325  HOME
326  PRINT "MEAN VEL. OF ";D$
327  GOSUB 420
328  PRINT "DEPTH" TAB( 10)"MV" TAB( 17)"SDV" TAB( 26)"MU" TAB( 33)"SDU"
329  FOR I = 1 TO 5
330 RD = MXV(I): GOSUB 410:MX = RD
331 RD = XSD(I): GOSUB 410:XS = RD
332 RD = MYV(I): GOSUB 410:MY = RD
333 RD = YSD(I): GOSUB 410:YS = RD
334  PRINT "  "Z(I) TAB( 10)MX; TAB( 17)XS; TAB( 26)MY; TAB( 33)YS
335  NEXT I
336  GET W$: GET W$
337  RETURN
338  REM


400  REM  ------------------------
401  REM  MISCELLANEOUS SUBROUTINES
402  REM  ------------------------

410  REM  SUB FORMAT
411 RD =  INT (RD * 100 + .5):RD = RD / 100
412  RETURN
413  REM
420  REM  SUB LINING
421  PRINT
422  FOR P = 1 TO 40
423  PRINT "-";
424  NEXT P: RETURN
425  REM
```

BIOGRAPHY

Mr. SUPICHAI TANGJAITRONG was born on May 15, 1958, in Bangkok, Thailand. He recieved a B.Sc. in Physics from the Faculty of Science, Chulalongkorn University in 1980.